

```

1 // Dave Barry, Francis Crick Institute
2 // 2018.01.17
3 // david.barry@crick.ac.uk
4
5 // Splits all files in input directory into constituent channels and saves results in separate sub-directories
6
7 macro "Batch Split"{
8     format = ".tif"; // specify image format
9
10    directory = getDirectory("Choose input files"); // get input directory
11    print("Input: " + directory);
12    fileList = getFileList(directory); // get file list
13    print(fileList.length + " files.");
14
15    outputDirectory = getDirectory("Choose output directory"); // get output directory
16    print("Output: " + outputDirectory);
17
18    setBatchMode(true); // supresses windows opening
19
20    for (i = 0; i < fileList.length; i++) {
21        file = directory + File.separator + fileList[i];
22        if(endsWith(file, format)){ // check if file is correct format
23            print("\nFile " + file + " is a recognised format - processing.");
24            run("Bio-Formats Importer", "open=[" + file + "] color_mode=Default rois_import=[ROI manager] view=Hyperstack"); // open image with bioformats
25            getDimensions(width, height, sizeC, slices, frames); // get image dimensions
26            print("Number of channels: " + sizeC);
27            if(sizeC > 1){
28                run("Split Channels"); // split the image into constituent channels
29            }
30            subDirs = newArray(sizeC);
31            for(c = 0; c < sizeC; c++){ // create subdirectories for each channel
32                subDirs[c] = outputDirectory + "C_" + (c + 1);
33                if(!File.exists(subDirs[c])){
34                    File.makeDirectory(subDirs[c]);
35                }
36            }
37            titles = getList("image.titles"); // get the list of open images i.e the individual channels
38            for(j = 0; j < titles.length; j++){
39                selectWindow(titles[j]);
40                filename = subDirs[j] + File.separator + titles[j] + "_C" + (j + 1) + ".ome.tif";
41                print("Saving " + filename);
42                run("Bio-Formats Exporter", "save=[" + filename + "] use compression=Uncompressed"); // save image
43                close();
44            }
45        } else {
46            print("\nFile " + file + " is not a recognised format - skipping.");
47        }
48    }
49    close("");
50
51    print("\nFinished");
52    showStatus("Finished.");
53    setBatchMode(false);
54 }

```